
Uz web-risinājumiem orientētās 

programmēšanas valodas un CGI-scripti.
Pēdējā laikā parādījās daudz interpretātoru, kuru galvenais mērķis ir nodrošināt ērtu un vienkāršu web-orientētu programmu (*CGI-skriptu) izstrādi. Visspilgtākais tādas valodas piemērs ir PHP. Pastāv arī vairākas valodas, kam ir plašs lietojumu loks, bet tās ir ļoti labi pielāgotas arī CGI-skriptu izstrādei, kā, piemēram, Perl. Tā kā aktīvā web satura daļa arvien pieaug (dinamisko web-lappušu skaits, t.i. lappušu, kas tika ģenerētas ar CGI-skriptu palīdzību), parādās arvien vairāk izstrādes līdzekļu, rīku un  programmēšanas valodu.ar to saistīto problēmu risinināšanai.

Šis referāts ir veltīts vienai no tādām valodām – valodai Parser.

Bet, pirms valodas Parser sintakses un semantikas īpatnību izklāstīšanas, ir jāapskata CGI-programmēšanas koncepcija kopumā.

Lietotāja pārlūkprogramma, vēršoties pie kāda statiska dokumenta web-serverī, faktiski liek dokumentu ar doto vārdu nodot atpakaļ lietotājam, neieviešot tajā nekādu izmaiņu. Šajā gadījumā, vienīgā darbība, ko veic web-serveris ir kontrolēt vai dotajam lietotājam ir tiesības izmantot viņa pieprasīto dokumentu, pārveidot URL adresi, izmantojot web-servera konfigurācijas noteikumus, lokālās failu sistēmas ceļā, u tm.l. sīkas, ne īpaši prasīgas pret procesora resursiem, darbības. Bet vairumu no pastāvošajiem web-severiem var uzkonfigurēt tā, lai vēršoties pie kādiem konktrētā tipa failiem (piemēram, pie visiem *.php failiem) tiktu izraisītas papilddarbības pirms rezultāta nodošanas lietotājam. Tādā gadījumā, dokuments jau pārtop par dinamisku. Mēs neaplūkosim sīki CGI iekšējos mehānismus, kā notiek dažādu parametru nodošana CGI-programmai, kādi ir vides mainīgie, kurus var izmantot savās CGI-programmās.  Mums dotajā brīdī ir svarīgi, ka pieprasot kādu konkrētu dokumentu, kas web-servera konfigurācijā ir speciāli norādīts, serveris nenodod to uzreiz lietotājam, bet izsauc konfigurācijā norādīto, visbiežāk, ārējo programmu (interpretātoru), ar kuras palīdzību tiek apstrādāts dotais dokuments, bet web-serveris atgriež lietotājam jau šīs izsauktās programmas rezultātus (standartizvadu). Ne vienmēr tiek izsauktas ārējas programmas, lai apstrādātu kādu noteiktu faila tipu. Pašlaik, daudziem web-serveriem pastāv iespēja kompilācijas laikā pievienot kādas valodas apstrādes moduli pašam web-serverim. Bez tam pastāv iespēja izpildīt tieši pieprasīto failu, kā dotās operētājsistēmas izpildāmo moduli (bieži, tie ir faili ar .cgi paplašinājumu). Shēmā 1 ir attēlotas lietotāja pārlūkprogrammas, web-servera un cgi-programmu mijiedarbība.

Parser valodas vēsture, interpretātora uzstādīšana

un vienkāršākās programmas.
Parser valoda ir Krievijā plaši pazīstamās Artemija Ļebedeva web-dizaina studijas proukts. Sākotnēji tā tika radīta, lai atvieglotu webmeisteram dažas darbietilpīgās operācijas. A. Ļebedeva studija šobrīd izplata šīs valodas 3. versiju  Tā ir brīvi pieejama ar izejas tekstiem C++ valodā un to var brīvi izmantot. Bet šis referāts ir veltīts valodas Parser 2. versijai (kas vēl joprojām ir brīvi pieejama, bet tikai tās izpildāmie moduļi, kompilēti priekš dažādām operētājsistēmām). Jo tieši ar šo valodas Parser versiju es esmu labāk pazīstams. Tā ir interesanta vēl ar to, ka tā tika izmantota iekšējām A.Ļebedeva studijas vajadzībām laikā, kad viņa studija, faktiski, tikai sāka iegūt popularitāti un vēl nebija nostiprinājusies IT tirgū. Un kaut arī Parser tehnoloģija atstāj dažreiz “mācību līdzekļa” nevis “darbarīka” iespaidu, tas nekļuva par šķērsli A.Ļebedeva studijai šājā laika posmā izpildīt projektus, kas pretendēja uz ļoti lielu nopietnību, viņu klienti bija vērā ņemamas kompānijas.

Kā jau tas tika minēts augstāk, valoda Parser ir domāta CGI-skriptu izveidošanai. Parser interpretātors var būt “piekabināts”, piemēram, Apache serverim, kā vienkārša ārēja CGI-programma (izpildāmam modulim ir jāatrodas /cgi-bin/ direktorijā), bet servera konfigurācijā ir jānorāda failu tipu, kas tiks apstrādāts ar Parser interpretātoru web-servera darbības laikā. Atšķirībā no citām tāda paša tipa valodām, kurā rakstītajiem skripta failiem ir jābūt ar kādu noteiktu paplašinājumu (piemēram .php), valodai Parser tradicionāli šis paplašinājums sakrīt ar vienkāršo HTML-failu paplašinājumu (.html), kaut gan tas nav obligātais nosacījums. Apache servera konfigurācijā tas izskatās sekojoši:

# ar šo rindiņu mēs pasakām, ka visi html faili tiek pakļauti apstrādei, 

# nevis nodoti lietotājam bez apstrādes

AddHandler parsed-html html

# ar šo rindu mēs pasakām, ka .html faili tiek nodoti apstrādei parser2.cgi 

# programmai, kas ir Parser valodas interpretātors.
Action parsed-html /cgi-bin/parser2.cgi


Parsera valodas koncepcija tiek atspoguļota nosaukumā: pēc būtības programmas teksts ir vienkāršs HTML-fails, kas arī tiek nodots lietotājam pēc apstrādes tādā pašā veidā, bez izmaiņām, izņemot konstrukcijas, kas pieder tieši Parser valodai. Kā tad tiek atšķirts HTML-teksts no Parser-programmas ? Gandrīz visas Parser komandas sākas ar simbolu ^ (tā kā tādu simbolu programmas tekstā ir samērā daudz, tas tika atspoguļots Parser logotipā, jo Krievijā šo simbolu sarunvalodā bieži dēvē par “putniņu”).

Lai nodemontrētu Parser valodas koncepciju apskatīsim vienkāršu piemēru, kas aprēķina faktoriāla vērtību dotajam n. Apskatot šo un turpmākos piemērus mēs pieņemsim, ka tie atrodas pēc adreses http://localhost/ saknes direktorijā.

[1]# Fails: factorials.html

[2]<html>

[3]<body>
[4] <h1>n=^number[n]</h1>
[5] ^var[fac;1]
[6] ^for[i;1;^number[n];^var[fac;*;^var[i]]]
[7] <h1>^number[n]!=^var[fac]</h1>
[8]</body>

[9]</html>
Pieņemot, ka mēs veicam pieprasījumu serverim: http://localhost/factorials.html?n=5 , mēs iegūsim sekojošu rezultātu:

n=5

5!=120

Tagad mēģināsim tikt skaidrībā ar to, ko tad nozīmē katra šīs programmas rindiņa. * 

Rindiņā [1] mēs redzam tikai komentāru, tas tiek ignorēts un netiek nosūtīts letotājam, tas darbojas kā komentārs tikai, ja pirmais simbols šajā rindiņā ir #, ja tā ir, piemēram, atstarpe, tad ši rindiņa tiek nodota apstrādei (kas starp citu ir ļoti neērti). Ja mums būtu nepieciešams izveidot komentāru, kas būtu garāks par vienu rindiņu, vai arī tas atrastos rindiņas vidū, tad būtu jāizmanto operātoru ^rem[], kura vienīgais arguments ir komentārs, kas tiek ignorēts.  Rindiņās  [2,3,8,9]  ir  tikai HTML-tegi. [4] rindiņā  teksts “ <h1>n=” un “</h1>” tiks nodoti lietotājam bez apstrādes. Turpretīm, ^number ir Parser predefinētais operātors, kura semantika ir sekojoša: mēs ņemam parametru, kas tika nodots CGI-skriptam vai nu ar POST (no formas), vai ar GET metodi (kā mūsu gadījumā, pierakstot parametrus pēc jautājuma zīmes URL-adresē), pārveido to par veselo nenegatīvo skaitli, ja izrādās, ka to izdarīt nevar, piemēram, ja parametrā bija teksts (n=qwerty), tad ^number[n] atgriež 0, t.i. tas darbojas kaut kādā ziņā līdzīgi, kā to dara funkcijas citās programmēšanas valodās, ar to nelielu atšķirību, ka dotajā gadījumā rezultāts tiek pievienots vienkārši visam pārējam tekstam, ar visiem HTML-tegiem. Valodā Parser pastāv arī ^form[<parametra vārds>] operātors, kas atgreiž parametrus nepārveidojot tos skaitļos. Tas, ka pastāv tāds operātors, kā ^number[] nav nejaušība. CGI-programmas parametros bieži nodod kaut kādu identifikātoru datubāzes tabulā, lai izmantotu to veicot kādu SQL-vaicājumu. Ja mēs izmantotu ^form[] operātoru, tad mēs varētu pieļaut web-lietotājam ļaunprātīgi SQL-vaicājumā izraisīt kļūdas, vai pat viņš varētu veikt kādas destruktīvās darbības. Kā redzam, izskaidrot kādas noteikta operātora īpatnības nemaz nav tik vienkārši. Kāpēc? Atbilde ir meklējama Pareser valodas pirmssākumos, jo tā tika veidota webmeistera darba atvieglošanai, lai ar īsām komandām tiktu pildītas daudzas sīkas, tipveida operācijas. 

Beidzot mēs nonācām pie [5] rindiņas. Tājā mēs nodefinējam mainīgo ar vārdu fac un piešķiram tam vērtību 1. No šīs rindiņas mēs varam izdarīt vairākus secinājumus: pirmkārt, mainīgie ir imperatīvās programmēšanas valodas pazīme un Parser nenoliedzami tāda ir, otrkārt, mainīgos nav iepriekš jādefinē, to definēšana sakrīt ar vērtības piešķiršanu, ja vērtība netika piešķirta, piemēram ^var[k;] (^var[] operātoram var būt divi argumenti, tad tas ir piešķiršanas operātors, argumenti atdalās ar semikolu), tad uzskata, ka mainīgais nav definēts (to vai izteiksme ir definēta var pārbaudīt ar operātoru ^ifdef[]). Izlaižot [6] rindiņu, pateiksim, ka [7] tiek izvesta parametra n skaitliskā vērtība un mainīgā fac vērtība (^var[] ar vienu argumentu izvada mainīgā vērtību). Lūk, mēs nonācām pie vissarežģītākās šīš programmas rindiņas – [6]: tajā mēs redzam ciklu. Cikli Parser valodā tiek realizēti diezgan īpatnējā stilā – tie līdzinās funkcijām. Rindiņā [6] mēs redzam ^for[] ciklu, tam ir četri parametri: pirmais cikla mainīgā vārds, otrais – skaitītāja sākumvērtība, trešais – skaitītāja galavērtība, bet ceturtais paramters – Parser koda gabals! Šo ciklu mēs izpildām n reižu. Cikla iekšienē ir ^var[fac;*;^var[i]] , kas ir operācija ar mainīgajiem, ko valodā C apraksta, kā fac*=i . Funkcijai ^var[] ar trijiem operandiem ir sekojoša semantika: jaunai fac vērtībai piešķirt veco fac vērtību izdarot operāciju no otrā argumenta, kā otru operācijas operandu izmantojot mainīgo no trešā ^var[] operātora argumenta. Kā  operāciju trīsargumentu ^var[] izsaukumos var izmantot +,-,*,/ (veseliem skaiļiem) vai . – simbolu virkņu konkatinācijai. Kā redzam Parser valodai nav stingras tipizācijas mainīgajiem, piemēram, piešķirot jauno vērtību mainīgam nav nosvara, vai mēs piešķiram skaitli, vai simbolu virkni, atšķirības rodas tikai veicot specifiskas katram tipam darbības.



Dažas Parser valodas funkcijas.

Pieņemot to, ka lasītāji ir pazīstami ar imperatīvās programmēšanas valodu konstrukcijām, es tikai pārskaitīšu tos analogus Parser valodā tabulas veidā, neiedziļinoties to semantikas skaidrošanā.

Zarošanas operātoru ekvivalenti

^eq[<a>;<b>;<c>;<d>]
Ja izteiksme <a> ir vienāda ar <b>, tad izpildām <c> Parser-koda bloku, citādi izpildām <d>.

^ifdef[<a>;<b>;<c>]
Ja izteiksmei <a> ir netukša vērtība, tad izpildām <b>, pretējā gadījumā <c>

^if[<a>;<b>;<c>]
Ja loģiskā izteiksme <a> ir spēkā, tad izpildām <b>, pretējā gadījumā <c>. Izteiksmē <a> var izmantot sekojošas operācijas: <, >, <=, =>, ==, !=, &&, ||, !

^switch[<izteiksme>;
 ^case[<vert_a>;<a>]
 ...
 ^case[<vert_b>;<z>]

 ^case[DEFAULT;<default>]
]
Pārbauda izteiksmi un ja tā ir vienāda ar kādu no ^case[] konstrukcijās norādītājām vērtībām, tad izpilda atbilstošu ^case[] konstrukcijā norādīto bloku, ja tāda vērtība netika atrasta, tad izpilda, ja tāda pastāv , ^switch[] konstrukcijas ar ^case[DEFAULT] saistīto koda bloku.

^cmp[<a>;<b>;<c>;<d>;<e>]
Veselo skaitļu salīdzināšanas operātors: <a> un <b> - veseli skaitļi; <c> izpildās, ja skaitlis <a> < <b>, <d>, ja <a> = <b> un <e>, ja <a> > <b>

Cilka operātori

^while[<a>;<b>]
<a> - loģiska izteiksme, sk. ^if[], <b>, kods, kas ir jāizpilda tik reizes, kamēr <a> ir spēkā, bet ne vairāk, kā 10 000 reižu; lielāks cikla izpildes skaits izraisīs kļūdu [endless loop detected].

^for[<a>;<b>;<c>;<d>;<e>]
<a> - cikla mainīgā vārds, <b> - cikla mainīgā sākumvērtība, 

<c> - cikla mainīgā gala vērtība, <d> -  cikla ķermenis, tas tiek izpildīts <c>-<b>+1 reižu, <e> - neobligātais parametrs, atdalītājs, kods, kas tiek izpildīts cikla beigās katru reizi, izņemot pēdējo (noder HTML tabulu veidošanai, u t.l. nolūkos, lai neveiktu lieku ^eq[] vai ^if[] pārbaudi)

Dabības ar mainīgajiem, aritmētiskās operācijas

^var[<a>]
Atgriež mainīgā <a> vērtību

^var[<a>;<b>]
Nodefinē mainīgo <a> un piešķir tam vērtību <b>, ja <b> ir tukšs, tad mainīgais <a> pārtop par nedefinētu

^var[<a>;<o>;<b>]
<a>=<a><o><b>, kur <o> ir +,-,*,/ veseliem skaitļiem un . simbolu virknēm.

^var[<a>;<f>;<b>]
Piešķir mainīgam <a> vērtību <b>, kas ir aritmētiskā izteiksme, izmantojot formatēšanas paraugu <f> (līdzīgu C printf sintaksei)

^eval[<a>;<f>]
Atgriež aritmētiskās izteiksmes <a> vērtību formatējot to ar <f>, <a> var saturēt aritmētiskā operācijas mainīgo vārdus (bez ^var[] izsaukumiem) un tabulas tekošās rindas elementu vārdus .

Valoda Parser ir modulāra. Tā ļauj lietotājiem definēt un izsaukt (ar ^macro[] operātoru) savus makrosus ar parametriem (līdzīgi, kā tas ir ar funkcijām citās programmēšanas valodās), tā ļauj izmantot ārējos moduļus (ar ^macro_use[]), pastāv iespējas katram web-projekta failam izpildīt vienādus makrosus apstrādes sākumā (speciālais makross @autoexec[]) un pastāv arī automātiski ielādējamais modulis projekta saknes katalogā _macro.cfg.  Tas var ievērojami atvieglot projekta veidošanu. Turklāt, interpretātora koncepcija ļauj pievienot un izpildīt darbības laikā atsevišķus Parser koda blokus, kuri ir paņemti no ārienes, kā dati, piemēram no datubāzes (ar ^process[] operātoru).

Parser valodā ir daudz īpatnēju konstrukciju, to sīka izklāstīšana, diemžēl, nevar būt iekļauta šajā referātā. Bet vienu no tām es tomēr mēģināšu aplūkot sīkāk. Tā ir tabulu koncepcija.
Tabulu koncepcija valodā Parser.

Par šo valodu, droši vien, nebūtu vērts runāt, ja tā ar kaut ko neatšķirtos no citām un, vēl vairāk,  kaut kādā ziņā nebūtu labāka par tām. Atcerēsimies vēlreiz, priekš kam tā tika radīta? Lai atvieglotu dzīvi webmeisteram, kam ir jāveic daudz tipveida sīku darbību, lai tās automatizētu – ne vairāk par to! Tāpēc valodai Parser, vismaz tās 2. versijai, dažreiz ir ļoti ierobežota funkcionalitāte. No otras puses, šajā valodā ir iespējas, kuras var realizēt ar vienu izsaukumu, turpretīm citās valodās, nāktos rakstīt savu simtiem rindiņu teksta garu bibliotēku. 

Viena no tādām īpatnējām Parser konstrukcijām ir darbs ar tabulām. Aplūkosim to  piemērā. Pieņemsim, ka ar tabulācijas simboliem  atdalītā  teksta  failā  mēs  glabājam  tabulu,  kas  sastāv no divām kolonām – cilvēku vārdiem un telefona numuriem:

Tagad mēs gribētu izveidot CGI-programmu, kas ļautu veikt meklēšanu šajā teksta failā vai nu pēc vārda, vai pēc telefona numura. Datus no teksta faila ielādēsim tabulā, kur arī veiksim meklēšanu.

# phonebook.html

@main[]

 ^ifdef[^form[pattern];

# ja ir definēts meklēšanas paraugs – sākam meklēšanu

  ^macro[make_search]

  ;

# pretējā gadījumā izvadām HTML formu, kur var ievadīt meklēšanas paraugu

  ^macro[draw_form]
 ]

# lietotāja nodefinētais makross, kas veic meklēšanu tabulā

@make_search[]

^context[phonebook;

# sākam darboties ar tabulu “phonebook”

 ^load[named;phonebook.txt]
# ielādējam datus tabulā no teksta faila _phonebook.txt

# teksta failu nosaukumiem ir jāsākas ar _ simbolu, ja mēs to ielādējam ar load[]

^var[field;^right[^form[search];^eval[^length[^form[search]]-2]]]
# mainīgajā ‘field’ atradīsies tā lauka vārds, pēc kura mēs veiksim meklēšanu

^locate[^var[field];^form[pattern]]

# meklējam nepieciešamo ierakstu noteiktajā tabulas kolonā

<h2>

^eq[^item[^var[field]];^form[pattern];

# izvadām tekošās rindas elementus, ja ieraksts tika atrasts

^item[name] - ^item[phone]
;

Diemžēl Jūsu meklētais ieraksts netika atrasts!]
]

</h2>

# lietotāja makross, kas izvada HTML-formu

@draw_form[]

^ifdef[^form[search];^var[s;^form[search]];^var[s;byname]]
<form method="POST">

<input type="text" size="20" name="pattern">

<input type="submit" value="Meklēt"><br>

<input type="radio" name="search" value="byname" ^eq[^var[s];byname;checked]> pēc vārda<br>
<input type="radio" name="search" value="byphone" ^eq[^var[s];byphone;checked]> pēc telefona numura<br>

</form>

Protams, darbs ar tabulām praksē ir daudz plašāks un interesantāks. Tabulu koncepcija ir savietojama ar SQL-pieprasījumu rezultātiem, t.i. ^sql[] operātora rezultāti SELECT-vaicājumu gadījumā tiek saglabāti tekošajā tabulā. Darbam ar tabulām Parser valodā paredzēti vairāk nekā 10 operātori (meklēšanai, navigācijai tabulas ietvaros, elementu kārtošanai un pilnai pārlasei).

Valodas Parser trūkumi un to novēršana

Valodā Parser ir ļoti daudz trūkumu, vismaz tās 2. versijā. Tajā nav daudzu vispārpieņemto konstrukciju un dažreiz tas traucē koncentrēties uz uzdevuma risināšanas. Bet, bieži, ja mēs paskatāmies uz pašu valodas Parser koncepciju, mēs varam atrast alternatīvos risinājumus, kuri nebūt nav sarežģīti un, dažreiz, pat izdomāt labāku risinājumu, kas būtu bāzēts tieši uz Parser valodas īpatnējām konstrukcijām.

Valodā Parser nav tāda jēdziena, kā masīvs. To gan bez īpašām grūtībām praksē var realizēt, jo Parser ir interpretātors un tāpēc dodot mainīgajiem vārdus var izmantot citus mainīgus. “Masīva” realizācija ir aplūkota sekojošā piemērā, kas ir pilnveidota 1. piemēra versija:

# factorial2.html: izveidojam “masīvu” A ar indeksiem no 1 līdz n un aizpildām to

# ar n pirmām faktariāla vērtībām Ai=i!, pēc tam izvadām rezultātu no “masīva”
@main[]

 ^macro[build_array;^number[n]]
 ^macro[print_array;^number[n]]
@build_array[count]

 ^var[fac;1]
 ^for[i;1;$count;^var[fac;*;^var[i]]^var[A^var[i];^var[fac]]]
@print_array[count]

 ^for[i;1;$count;^var[i]!=^var[A^var[i]];<br>]

Veicot pieprasījumu web-serverim http://localhost/fibonacci.html?n=5, rezultātā iegūsim:

1!=1
2!=2
3!=6
4!=24
5!=120

Vēl darbu ar masīvu gandrīz vienmēr var reducēt uz darbu ar tabulām.

Kā vēl vienu būtisku trūkumu varētu minēt to, ka Parser valodā nav rekursijas. Tomēr, izrādās, ka arī šo ierobežojumu var viegli apiet, dinamiski piešķirot lietotāja makrosiem jaunus vārdus, kur beigās ir skaitlis, kas atspoguļo tekošo rekursijas dziļumu.

Parser valodā ir ļoti nepatīkams atkļūdošanas process, bieži vien tas ir saistīts ar primitīvo sintakses kļūdu meklēšanu. Liels daudzums kvadrātiekavu neļauj apieties bez speciāliem teksta redaktoriem, kas palīdz atrast pāra iekavas (piemēram, Midnight Commander, iebūvētais redaktors). Nepatīkami ir tas, ka interpretātora reakcija uz kļūdām var arī neizraisīt kļūdas paziņojumu, kas padara kļūdas meklēšanas procesu ievērojami sarežģītāku.

Tomēr ir arī nopietnāki trūkumi, precīzāk pat būtu nosaukt tos par kļūdām Parser interpretātorā. Piemēram, ja mēs izveidojam lietotāja makrosu un tajā sākam darboties ar kādu jaunu tabulu (atveram konekstu), mēs nevaram piekļūt pie makrosa parametriem, tie paliek par nedefinētiem. Saskarsme ar tāda tipa kļūdām pašā interpretātorā izraisa ļoti nepatīkamas sajūtas un dažreiz ievērojami samazina atkļūdošanas ātrumu, jo programmētājs, tomēr, pierada meklēt kļūdas programmas loģikā nevis to izpildošā interpretātorā.

Vēl viens nopietns trūkums ir Parser valodas 3. versijas nesavietojamība ar 2. versiju. Kaut gan pastāv speciālā programma-konvertors (Pt2toPt3), tas var sagādāt daudz nepatikšanu tiem, kuru web-projekti ir veidoti ar Parser 2. versiju. Tomēr ir jāsaka, ka būtiski izmainot dažas valodas koncepcijas 3. versijā (tur pamatā ir objekt-orientētā pieeja), izstrādātāji novērsa arī vairākas 2. versijas nepilnības un izlaboja kļūdas. Turklāt 3. verisja ir brīvi pieejama ar interpretātora izejas tekstiem, tā kā pamanot kādu kļūdu pašā interpretātorā, var to viegli izlabot. 

Rezumējot, gribētos pateikt, ka kopumā Parser valodas vismaz 2. versija tomēr nav tik elastīga un ērta, kā gribētos, tāpēc, pielietojot to praksē, varētu sagādāt sev liekas grūtības, kuras varētu pārvarēt izvēloties kādu citu līdzīgu programmēšanas valodu, kurai būtu plašāka funkcionalitāte, bet kura tomēr būtu labi pielāgota CGI-programmēšanas specifikai.

Izmantotie resursi

1. Valodas Parser 2.versijas oficiālā mājas lapa – http://parser2.parser.ru
Dokumentācija un lietotāju forums
2. Valoadas Parser 3. versijas oficiālā mājas lapa – http://www.parser.ru
Valodas dokumentācija.

# _phonebook.txt


name		phone


Aleksandrs	7123456


Dace		1234567


. . .


Liene		2567123


Ivars		7135634





* CGI – Common Gateway Interface, protokols, ar kura palīdzību lietotājprogrammas un web-serveris var mijiedarboties, veidojot dinamiskās web-lappuses; pastāv arī citi līdzīgi protokoli





* Šis un visi pārējie piemēri tika rakstīti un pārbaudīti uz valodas Parser 2. versijas interpretātora zem Apache 1.3 web-servera (win32-versijas)








